User-defined functions

In[ ]:=
```
KroneckerSum[A_, B_] := Module[
  {m, n},
  m = Length[A];
  n = Length[Transpose[A]];
  KroneckerProduct[A, IdentityMatrix[m]] + KroneckerProduct[ IdentityMatrix[n], B]]
```

In[ ]:=
```
(* Vec operator *)
Vec[A_] := Flatten[A, {2, 1}]
```

In[ ]:=
```
(* covariances of the stationary distribution of a system of stochastic differential
 equations (continuous-time). Assumes all matrixes have dimensions m X m*)
StatCov[A_, B_, C_] := Module[
  {m, covVec},
  m = Length[A];
  covVec = -Inverse[KroneckerSum[A, A]].Vec[B.C.Transpose[B]] // Simplify;
  Transpose[ArrayReshape[covVec, {m, m}]]]
```

In[ ]:=
```
(* covariances of the stationary distribution of a system of stochastic difference
 equations (discrete-time). Assumes all matrixes have dimensions m X m*)
StatCovDiscrete[A_, B_, C_] := Module[
  {m, covVec},
  m = Length[A];
  covVec = Inverse[IdentityMatrix[m^2] - KroneckerProduct[A, A]].
     Vec[B.C.Transpose[B]] // Simplify;
  Transpose[ArrayReshape[covVec, {m, m}]]]
```

EC Covariance for a resident species

In[ ]:=
```
(* define the drift and diffusion matrixes *)
A = {{F"N", F"E"}, {0, G"E"}};
B = {{0, 0}, {0, σ}};
WCor = {{1, 0}, {0, 1}}; (* wCor is C in the main text,
but C is a protected symbol in Mathematica *)
```

In[ ]:=
```
(* the covariances of the stationary distribution *)
statCov = StatCov[A, B, WCor];
statCov // MatrixForm // Simplify
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{\sigma^2 F_E^2}{2 F_N^2 G_E + 2 F_N G_E^2} & \frac{\sigma^2 F_E}{2 F_N G_E + 2 G_E^2} \\ \frac{\sigma^2 F_E}{2 F_N G_E + 2 G_E^2} & -\frac{\sigma^2}{2 G_E} \end{pmatrix}$$

In[ ]:=
```
(* relationships between ordinary model parameters and the time-scale parameters *)
```

$$\text{subs} = \left\{G_{\text{"E"}} \to -\frac{1}{T_{\text{"E"}}},\ \tilde{C}_{\text{"N"}} \to \frac{1}{F_{\text{"E"}}\, T_{\text{"E}\to\text{C"}}}\right\};$$

In[∘]:=
```
(* express Cov(E,C) in terms of the timescale parameters *)
covEC = C̃"N" statCov[[1, 2]] /. subs // FullSimplify
```

Out[∘]=

$$\frac{\sigma^2\, T_E^2}{2\, T_{E\to C} - 2\, F_N\, T_E\, T_{E\to C}}$$

In[∘]:=
```
(* shows that Cov(E,C) increases with T"E" and decreases wtih T"E→C". The
 reduce function takes a sequence of assumptions (separated by "&&")
 and reduces them in number. The fact that the additional assumption
 (i.e., that the derivative of covEC with respect to T"E" is positive) does not
 show up in the output of reduce means that this is a redundant assumption -- it
 could be derived from the constraints on the elements of the A and B matrixes,
which are captured in the variable named "$Assumptions" *)
$Assumptions = σ > 0 && F"N" < 0 && T"E" > 0 && T"E→C" > 0;
Reduce[$Assumptions && D[covEC, T"E"] > 0]
Reduce[$Assumptions && D[covEC, T"E→C"] < 0]
```

Out[∘]= $F_N < 0\ \&\&\ T_E > 0\ \&\&\ T_{E\to C} > 0\ \&\&\ \sigma > 0$

Out[∘]= $F_N < 0\ \&\&\ T_E > 0\ \&\&\ T_{E\to C} > 0\ \&\&\ \sigma > 0$

EC Covariance for an invader

In[∘]:=
```
(* define the drift and diffusion matrixes *)
A = {{"F_{N_r}^{(r)}", "F_{E_r}^{(r)}", 0},
    {0, "G_{E_r}^{(r)}", 0},
    {0, 0, "G_{E_i}^{(i)}"}};
B = {{0, 0, 0},
    {0, σr, 0},
    {0, 0, σi}};
WCor = {{1, 0, 0},
    {0, 1, ρ},
    {0, ρ, 1}}; (* wCor is C in the main text,
but C is a protected symbol in Mathematica *)
```

In[∘]:=
```
statCov = StatCov[A, B, WCor];
statCov // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{F_{E_r}^{(r)\,2}\,\sigma_r^2}{2\,F_{N_r}^{(r)\,2}\,G_{E_r}^{(r)}+2\,F_{N_r}^{(r)}\,G_{E_r}^{(r)\,2}} & \frac{F_{E_r}^{(r)}\,\sigma_r^2}{2\,F_{N_r}^{(r)}\,G_{E_r}^{(r)}+2\,G_{E_r}^{(r)\,2}} & \frac{F_{E_r}^{(r)}\,\rho\,\sigma_i\,\sigma_r}{\left(F_{N_r}^{(r)}+G_{E_i}^{(i)}\right)\left(G_{E_i}^{(i)}+G_{E_r}^{(r)}\right)} \\ \frac{F_{E_r}^{(r)}\,\sigma_r^2}{2\,F_{N_r}^{(r)}\,G_{E_r}^{(r)}+2\,G_{E_r}^{(r)\,2}} & -\frac{\sigma_r^2}{2\,G_{E_r}^{(r)}} & -\frac{\rho\,\sigma_i\,\sigma_r}{G_{E_i}^{(i)}+G_{E_r}^{(r)}} \\ \frac{F_{E_r}^{(r)}\,\rho\,\sigma_i\,\sigma_r}{\left(F_{N_r}^{(r)}+G_{E_i}^{(i)}\right)\left(G_{E_i}^{(i)}+G_{E_r}^{(r)}\right)} & -\frac{\rho\,\sigma_i\,\sigma_r}{G_{E_i}^{(i)}+G_{E_r}^{(r)}} & -\frac{\sigma_i^2}{2\,G_{E_i}^{(i)}} \end{pmatrix}$$

In[∘]:=
```
(* covariance between E_i and C *)
covEC = C̃"N" statCov[[3, 1]] // Simplify
```

Out[∘]=

$$\frac{F_{E_r}^{(r)}\,\rho\,\sigma_i\,\sigma_r\,\tilde{C}_N}{\left(F_{N_r}^{(r)}+G_{E_i}^{(i)}\right)\left(G_{E_i}^{(i)}+G_{E_r}^{(r)}\right)}$$

Seasonality

In[•]:=
```
(* notation differs slightly from the appendix,
in that population density is n instead of N *)
sol = DSolve[{n'[t] == F"N" (n[t]) + F"E" σ Sin[2 * Pi * t / p]},
    n[t], t][[1]] // FullSimplify
```

Out[•]= $\left\{n[t] \to e^{t F_N} c_1 - \frac{p\,\sigma F_E \left(2\pi \operatorname{Cos}\left[\frac{2\pi t}{p}\right] + p \operatorname{Sin}\left[\frac{2\pi t}{p}\right] F_N\right)}{4\pi^2 + p^2 F_N^2}\right\}$

In[•]:=
```
(* asymptotic solution (t→∞) of population density*)
nAsymp = n[t] /. sol /. c1 → 0
```

Out[•]= $-\frac{p\,\sigma F_E \left(2\pi \operatorname{Cos}\left[\frac{2\pi t}{p}\right] + p \operatorname{Sin}\left[\frac{2\pi t}{p}\right] F_N\right)}{4\pi^2 + p^2 F_N^2}$

In[•]:=
```
(* Cov(E,N) *)
covEN = (1 / p) Integrate[nAsymp * (σ Sin[2 * Pi * t / p]), {t, 0, p}] // FullSimplify
```

Out[•]= $-\frac{p^2 \sigma^2 F_E F_N}{8\pi^2 + 2 p^2 F_N^2}$

In[•]:=
```
(* relationships between ordinary model parameters and the time-scale parameters *)
subs = {p → T"E", C̃"N" → 1 / (F"E" T"E→C")};
```

In[•]:=
```
(* express Cov(E,C) in terms of the timescale parameters *)
covEC = C̃"N" covEN /. subs // FullSimplify
```

Out[•]= $-\frac{\sigma^2 F_N T_E^2}{8\pi^2 T_{E\to C} + 2 F_N^2 T_E^2 T_{E\to C}}$

In[•]:=
```
(* shows that Cov(E,C) increases with T"E" and decreases wtih T"E→C". The
 reduce function takes a sequence of assumptions (separated by "&&")
 and reduces them in number. The fact that the additional assumption
 (i.e., that the derivative of covEC with respect to T"E" is positive) does not
 show up in the output of reduce means that this is a redundant assumption -- it
 could be derived from the constraints on the elements of the A and B matrixes,
which are captured in the variable named "$Assumptions" *)
$Assumptions = σ > 0 && F"N" < 0 && T"E" > 0 && T"E→C" > 0;
Reduce[$Assumptions && D[covEC, T"E"] > 0]
Reduce[$Assumptions && D[covEC, T"E→C"] < 0]
```

Out[•]= $F_N < 0 \;\&\&\; T_E > 0 \;\&\&\; T_{E\to C} > 0 \;\&\&\; \sigma > 0$

Out[•]= $F_N < 0 \;\&\&\; T_E > 0 \;\&\&\; T_{E\to C} > 0 \;\&\&\; \sigma > 0$

Explicit resource competition

In[ ]:=
```
(* define the drift and diffusion matrixes *)
A = {{0, F"E", F"R"},
    {0, G"E", 0},
    {H"N", H"E", H"R"}};
B = {{0, 0, 0},
    {0, σ, 0},
    {0, 0, 0}};
WCor = {
    {1, 0, 0},
    {0, 1, 0},
    {0, 0, 1}}; (* wCor is C in the main text,
but C is a protected symbol in Mathematica *)
```

In[ ]:=
```
(* The covariances of the stationary distribution *)
statCov = StatCov[A, B, WCor];
statCov // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{\sigma^2\left(F_R^2 H_E^2 (G_E+H_R)+F_E^2 H_R^2 (G_E+H_R)-F_E F_R (F_E G_E H_N+2 H_E H_R (G_E+H_R))\right)}{2 F_R G_E H_N H_R \left(-G_E^2+F_R H_N-G_E H_R\right)} & \frac{\sigma^2 (-F_R H_E+F_E (G_E+H_R))}{2 G_E \left(G_E^2-F_R H_N+G_E H_R\right)} & \frac{\sigma^2 F_E (-F_R H_E+F_E (G_E+H_R))}{2 F_R G_E \left(-G_E^2+F_R H_N-G_E H_R\right)} \\ \frac{\sigma^2 (-F_R H_E+F_E (G_E+H_R))}{2 G_E \left(G_E^2-F_R H_N+G_E H_R\right)} & -\frac{\sigma^2}{2 G_E} & \frac{\sigma^2 (G_E H_E-F_E H_N)}{2 G_E \left(G_E^2-F_R H_N+G_E H_R\right)} \\ \frac{\sigma^2 F_E (-F_R H_E+F_E (G_E+H_R))}{2 F_R G_E \left(-G_E^2+F_R H_N-G_E H_R\right)} & \frac{\sigma^2 (G_E H_E-F_E H_N)}{2 G_E \left(G_E^2-F_R H_N+G_E H_R\right)} & \frac{\sigma^2 \left(F_R G_E H_E^2-F_E^2 H_N (G_E+H_R)\right)}{2 F_R G_E H_R \left(-G_E^2+F_R H_N-G_E H_R\right)} \end{pmatrix}$$

In[ ]:=
```
(* the eigenvalues of the matrix A *)
eA = Eigenvalues[A]
```

Out[ ]=
$$\left\{G_E, \frac{1}{2}\left(H_R-\sqrt{4 F_R H_N+H_R^2}\right), \frac{1}{2}\left(H_R+\sqrt{4 F_R H_N+H_R^2}\right)\right\}$$

In[ ]:=
```
(* Define known constraints on the model parameters *)
$Assumptions =
  σ > 0 && F"R" > 0 && F"E" > 0 && H"N" < 0 && H"R" < 0 && H"E" ≤ 0 && G"E" < 0 && C̃"N" < 0;
(* Here, we learn what additional constraints are needed in order for the
 deterministic system to be stable, which is equivalent to all eigenvalues < 0 *)
$Assumptions = Reduce[$Assumptions && And @@ Map[# < 0 &, eA]]
(* note the derived constraint: -H"R"^2/(4 F"R") ≤ H"N" < 0 *)
```

Out[ ]=
$$G_E < 0 \,\&\&\, \tilde{C}_N < 0 \,\&\&\, \sigma > 0 \,\&\&\, F_E > 0 \,\&\&\, F_R > 0 \,\&\&\, H_R < 0 \,\&\&\, -\frac{H_R^2}{4 F_R} \le H_N < 0 \,\&\&\, H_E \le 0$$

In[ ]:=
```
(* Cov(E,C) *)
covEC = C̃"N" statCov[[2, 3]] // Simplify
```

Out[ ]=
$$\frac{\sigma^2 (G_E H_E - F_E H_N)\, \tilde{C}_N}{2 G_E \left(G_E^2 - F_R H_N + G_E H_R\right)}$$

In[•]:= (* Cov(E,C) *)

```
covEC = C̃"N" statCov[[2, 3]] // Simplify
```

Out[•]= $$\frac{\sigma^2\,(G_E H_E - F_E H_N)\,\tilde{C}_N}{2\,G_E\,(G_E^2 - F_R H_N + G_E H_R)}$$

In[•]:= (* relationships between ordinary model parameters and the time-scale parameters *)

$$\text{subs} = \left\{G_{\text{"E"}} \to -\frac{1}{T_{\text{"E"}}},\ \tilde{C}_{\text{"N"}} \to \frac{1}{H_{\text{"N"}}\,F_{\text{"E"}}\,T_{\text{"E}\to\text{C"}}}\right\};$$

In[•]:= (* Cov(E,C), reparameterized. Note that the expression still contains $H_{\text{"N"}}$, which is a constituent of $T_{\text{"E}\to\text{C"}}$ *)

```
covECv2 = covEC /. subs // Simplify
```

Out[•]= $$-\frac{\sigma^2\,T_E^2\,(H_E + F_E H_N T_E)}{2\,F_E H_N\,(-1 + H_R T_E + F_R H_N T_E^2)\,T_{E\to C}}$$

(* Here we show that Cov(E,C) increases with $T_{\text{"E"}}$ and decreases wtih $T_{\text{"E}\to\text{C"}}$. We do not take the derivative of covEC with respect to the time-scale parameters, since the constituents of $T_{\text{"E}\to\text{C"}}$ cannot be fully eliminated from the expression for cov(E,C) (see covECv2) . Put another way taking the derivative of covECv2 with respect to $T_{\text{"E}\to\text{C"}}$ is like asking what happens to covECv2 when $T_{\text{"E}\to\text{C"}}$ increases but all other parameters stay the same; but this is nonsensical, because $T_{\text{"E}\to\text{C"}}$ increasing can be caused by $H_{\text{"N"}}$ decreasing. Instead, we look at the relationship between covEC and the constituents of $T_{\text{"E"}}$ and $T_{\text{"E}\to\text{C"}}$: $\tilde{C}_{\text{"N"}}$, $H_{\text{"N"}}$, $F_{\text{"E"}}$, $H_{\text{"E"}}$. As explained in the text of the appendix, when $H_{\text{"E"}}$=0, we expect that covEC is a decreasing function of $C_{\text{"R"}}$ and $H_{\text{"N"}}$; and that covEC is an increasing function of $F_{\text{"E"}}$ adn $G_{\text{"E"}}$. This turns out to be true. *)

In[•]:=

```
Reduce[$Assumptions && H"E" == 0 && D[covEC, C̃"N"] < 0]
Reduce[$Assumptions && H"E" == 0 && D[covEC, H"N"] < 0]
Reduce[$Assumptions && H"E" == 0 && D[covEC, F"E"] > 0]
Reduce[$Assumptions && H"E" == 0 && D[covEC,  G"E"] > 0]
```

Out[•]= $\tilde{C}_N < 0\ \&\&\ F_R > 0\ \&\&\ H_R < 0\ \&\&\ -\frac{H_R^2}{4\,F_R} \le H_N < 0\ \&\&\ F_E > 0\ \&\&\ G_E < 0\ \&\&\ \sigma > 0\ \&\&\ H_E == 0$

Out[•]= $F_R > 0\ \&\&\ H_R < 0\ \&\&\ -\frac{H_R^2}{4\,F_R} \le H_N < 0\ \&\&\ F_E > 0\ \&\&\ G_E < 0\ \&\&\ \tilde{C}_N < 0\ \&\&\ \sigma > 0\ \&\&\ H_E == 0$

Out[•]= $F_E > 0\ \&\&\ H_E == 0\ \&\&\ F_R > 0\ \&\&\ H_R < 0\ \&\&\ -\frac{H_R^2}{4\,F_R} \le H_N < 0\ \&\&\ G_E < 0\ \&\&\ \tilde{C}_N < 0\ \&\&\ \sigma > 0$

Out[•]= $F_R > 0\ \&\&\ H_R < 0\ \&\&\ -\frac{H_R^2}{4\,F_R} \le H_N < 0\ \&\&\ F_E > 0\ \&\&\ G_E < 0\ \&\&\ \tilde{C}_N < 0\ \&\&\ \sigma > 0\ \&\&\ H_E == 0$

In[∘]:=
```
(* As explained in the text of the appendix, when H"E"< 0,
we expect that covEC is a decreasing function of C"R" and H"E";
and that covEC is an increasing function of G"E". This turns out to be true. *)
Reduce[$Assumptions && H"E" < 0 && D[covEC, C̃"N"] < 0]
Reduce[$Assumptions && H"E" < 0 && D[covEC, H"E"] < 0]
Reduce[$Assumptions && H"E" < 0 && D[covEC, G"E"] > 0]
```

Out[∘]= $\tilde{C}_N < 0 \&\& F_R > 0 \&\& H_R < 0 \&\& -\frac{H_R^2}{4 F_R} \le H_N < 0 \&\& F_E > 0 \&\& G_E < 0 \&\& H_E < 0 \&\& \sigma > 0$

Out[∘]= $F_E > 0 \&\& H_E < 0 \&\& F_R > 0 \&\& H_R < 0 \&\& -\frac{H_R^2}{4 F_R} \le H_N < 0 \&\& G_E < 0 \&\& \tilde{C}_N < 0 \&\& \sigma > 0$

Out[∘]= $F_R > 0 \&\& H_R < 0 \&\& -\frac{H_R^2}{4 F_R} \le H_N < 0 \&\& F_E > 0 \&\& G_E < 0 \&\& H_E < 0 \&\& \tilde{C}_N < 0 \&\& \sigma > 0$

EC Covariance for a resident species; Discrete time

In[∘]:=
```
(* define the drift and diffusion matrixes *)
A = {{F"N", F"E"}, {0, G"E"}};
B = {{0, 0}, {0, σ}};
WCor = {{1, 0}, {0, 1}}; (* wCor is C in the main text,
but C is a protected symbol in Mathematica *)
```

In[∘]:=
```
(* the covariances of the stationary distribution *)
statCov = StatCovDiscrete[A, B, WCor];
statCov // MatrixForm // Simplify
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{\sigma^2 F_E^2 (1+F_N G_E)}{(-1+F_N^2)(-1+F_N G_E)(-1+G_E^2)} & \frac{\sigma^2 F_E G_E}{(-1+F_N G_E)(-1+G_E^2)} \\ \frac{\sigma^2 F_E G_E}{(-1+F_N G_E)(-1+G_E^2)} & \frac{\sigma^2}{1-G_E^2} \end{pmatrix}$$

In[∘]:=
```
(* Cov(E,C) *)
covEC = C̃"N" statCov[[1, 2]] // Simplify
```

Out[∘]= $$\frac{\sigma^2 F_E G_E \tilde{C}_N}{(-1 + F_N G_E)(-1 + G_E^2)}$$

Phytoplankton model

In[∘]:=
```
(* Dynamical equations. Note "n" and "e" are the
 population density and the environmental parameter respectively
 ("N" "E" are protected symbols in Mathematica). Similarly, "d" is the dilution rate *)
dndt := n (c Exp[e] R - δ)
dRdt := δ (S - R) - Exp[e] R n
dedt := -θ (e - μ) (* deterministic skeleton of the environmental parameter's dynamics *)
```

In[∘]:=
```
(* the single positive equilibrium *)
eq = Solve[{dndt/n == 0, dRdt/R == 0 , dedt/e == 0}, {n, R, e}][[1]]
```

Out[∘]= $\left\{n \to e^{-\mu}\left(c\, e^{\mu} S - \delta\right),\ R \to \frac{e^{-\mu}\delta}{c},\ e \to \mu\right\}$

In[∘]:=
```
(* Here, we learn what additional constraints are needed in
 order for the deterministic system to have a positive equilbirium *)
$Assumptions = σ > 0 && n > 0 && R > 0 && c > 0 && S > 0 && δ > 0 && μ ∈ Reals && θ > 0;
(* the last constraint is the requirement for the phytoplankton species
 to have a positive per capita growth rate with resources are abundant *)
$Assumptions = Reduce[$Assumptions && (n /. eq) > 0 && (R /. eq) > 0]
(* note the learned constraint μ>Log[(K δ+S δ)/(c S)] *)
```

Out[∘]= $\sigma > 0\ \&\&\ n > 0\ \&\&\ R > 0\ \&\&\ \theta > 0\ \&\&\ c > 0\ \&\&\ S > 0\ \&\&\ \delta > 0\ \&\&\ \mu > -\mathrm{Log}\left[\frac{c\,S}{\delta}\right]$

In[∘]:=
```
(* define the drift and diffusion matrixes *)
A = D[{dndt, dedt, dRdt}, {{n, e, R}}] /. eq;
B = {{0, 0, 0}, {0, σ, 0}, {0, 0, 0}};
WCor = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}; (* wCor is C in the main text,
but C is a protected symbol in Mathematica *)
A // FullSimplify // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & \delta\left(c\,S - e^{-\mu}\delta\right) & c\left(c\,e^{\mu}S - \delta\right) \\ 0 & -\theta & 0 \\ -\frac{\delta}{c} & \delta\left(-S + \frac{e^{-\mu}\delta}{c}\right) & -c\,e^{\mu}S \end{pmatrix}$$

In[∘]:=
```
(* covariances of the stationary distribution *)
statCov = StatCov[A, B, WCor];
statCov // MatrixForm // Simplify
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \frac{e^{-2\mu}\left(c\,e^{\mu}S-\delta\right)\delta^2\sigma^2}{2\theta\left(c\,e^{\mu}S-\delta+\theta\right)} & \frac{e^{-\mu}\left(c\,e^{\mu}S-\delta\right)\delta\,\sigma^2}{2\theta\left(c\,e^{\mu}S-\delta+\theta\right)} & -\frac{e^{-2\mu}\left(c\,e^{\mu}S-\delta\right)\delta^2\sigma^2}{2c\,\theta\left(c\,e^{\mu}S-\delta+\theta\right)} \\ \frac{e^{-\mu}\left(c\,e^{\mu}S-\delta\right)\delta\,\sigma^2}{2\theta\left(c\,e^{\mu}S-\delta+\theta\right)} & \frac{\sigma^2}{2\theta} & -\frac{e^{-\mu}\left(c\,e^{\mu}S-\delta\right)\delta\,\sigma^2}{2c\,\theta\left(c\,e^{\mu}S-\delta+\theta\right)} \\ -\frac{e^{-2\mu}\left(c\,e^{\mu}S-\delta\right)\delta^2\sigma^2}{2c\,\theta\left(c\,e^{\mu}S-\delta+\theta\right)} & -\frac{e^{-\mu}\left(c\,e^{\mu}S-\delta\right)\delta\,\sigma^2}{2c\,\theta\left(c\,e^{\mu}S-\delta+\theta\right)} & \frac{e^{-2\mu}\left(c\,e^{\mu}S-\delta\right)\delta^2\sigma^2}{2c^2\theta\left(c\,e^{\mu}S-\delta+\theta\right)} \end{pmatrix}$$

In[∘]:=
```
(* define the competition parameter *)
comp = Log[1/(c R)];
C̃"R" = D[comp, R] /. eq // Simplify
```

Out[∘]= $-\frac{c\,e^{\mu}}{\delta}$

In[∘]:=
```
(* Cov(E,C) *)
covEC = C̃"R" statCov[[2, 3]] // Simplify
```

Out[∘]= $\frac{\left(c\,e^{\mu}S - \delta\right)\sigma^2}{2\theta\left(c\,e^{\mu}S - \delta + \theta\right)}$

In[•]:=
```
(*T"E→C"*)
1/(C̃"R" * A[[3, 2]]) /. eq // FullSimplify
```

Out[•]=
$$\frac{1}{c\, e^{\mu}\, S - \delta}$$

Lottery Model (TBD)